
MEDIATION IN REGRESSION: 

WORK NOTES AND SYNTAX version 4
Winnifred R. Louis, School of Psychology, University of Queensland

w.louis@psy.uq.edu.au
© Winnifred Louis 2009
You can distribute the following freely for non-commercial use provided you retain the credit to me and periodically send me appreciative e-mails.

(Appreciative e-mails are useful for promotion & tenure, eh!  Not to mention gratifying.)
READER BEWARE - undergrads should read with caution - sometimes the advice re writing and analysis here contradicts what is advised in your courses.  Obviously you must follow the advice given in courses.  The discrepancies could be because 1) undergrad stats is an idealized version of reality whereas postgrads grapple with real data and publication pressure 2) statistical decision-making requires personal choices and different profs & practitioners may differ.
A wise practice for any write-up is to scan the intended publication outlet (the journal, other theses in the same lab, etc.) and try to find 3 or 4 examples of how the analysis has been written up before there.

What is a mediator?

It is a process or means by which the IV impacts on the DV.

The IV (self-esteem) impacts on grades (the DV) via motivation to study.

Hard drugs lead to increased mortality via increased risk taking.

Communication leads to relationship satisfaction via decreased misunderstanding.

In mediation, the IV and the mediator are associated (correlated), and the IV and the DV are correlated, and there is an implied causal path (“because”) that links the three variables.  The IV causes the DV because the IV causes the mediator which causes the DV.

One normally infers causality based on theory, NB.  The regression analyses show inter-relationships, but you need longitudinal analyses and/or experimental manipulations to show causation from design.

Mediators vs Moderators

To restate: In mediation, the IV and the mediator are associated (correlated), and the IV and the DV are correlated, and there is an implied causal path (“because”) that links the three variables.  The IV causes the DV because the IV causes the mediator which causes the DV.

In moderation (to get a significant interaction), the IVs need not be correlated with each other or with the DV.  In moderation, the link between the IV and the DV is different for high vs low levels of the moderator.  There is no because.  It’s more like if-then contingencies: If there’s high moderator, then the IV does this with the DV, and if there’s low moderator, the IV does this with the DV.

The IV (self-esteem) impacts on grades (the DV) but it’s moderated by motivation to study.  [At high motivation, there’s a link between self-esteem and grades, but at low motivation, there’s no link – everyone does badly.]

Hard drugs lead to increased mortality but it’s moderated by car ownership.

[At low car ownership, drugs lead to mortality, but at high car ownership the link is stronger.]

Communication leads to relationship satisfaction but it’s moderated by utterance valence. [If valence is positive, communication increases relationship satisfaction.  If negative, communication reduces relationship satisfaction.]

Writing up mediation in regression


A write-up for mediation in regression typically has four parts: the text, two tables, and a figure.  

· The figure shows the direct and indirect effects of the IV.  It would generally be included in theses; frequently dropped from manuscript to save space.
· Table 1 shows the uncentered means of all IVs and DVs, the standard deviations, and the inter-correlations among the variables.

· Table 2 shows the beta coefficients for each IV for each DV (or you can have separate tables for each DV).  Unlike the normal regression table, which usually shows only entry or final regression coefficients, the table with mediation usually shows the coefficients for each block, along with R2 change for each block and the final model R2.

· The write-up begins with an overview paragraph under the heading ‘design’ or ‘overview’ describing the analysis, centering, coding, treatment of missing variables and outliers, and zero-order correlations, and referring the reader to Tables 1 and 2.  NB you don’t have to center as long as you’re not creating interactions, but it’s usually a good idea.  Then there are often separate sections of the results for each DV.  Within the sections separate paragraphs or blocks of sentence describe each block.  It is noted whether adding the variables in each block increased the variance accounted for, and R2 ch and F statistics are given.  Some comment is made about the coefficients in each block.  Usually betas and p-values are reported.  

Mediation analysis requires one to report that the IV predicts the mediator and the DV, that the mediator predicts the DV, and that the link between the IV and the DV decreases when the mediator is controlled.

(1) To show the link between the IV and the mediator, you can refer the reader to Table 1 (a significant zero-order correlation), or conduct a separate regression analysis in which the IV (and any control variables) predict the mediator, reporting the total R2 and the significant beta for the IV.
(2) Then you run a hierarchical multiple regression analysis where the IV (and any control variables) predict the DV in Block 1, reporting the R2 ch and associated F, and the betas; and adding the mediator in Block 2, reporting the R2 ch and associated F, the beta for the mediator, and the new (smaller) beta for the IV.  The coefficient for the mediator should be significant and the coefficient for the IV should decrease from the original block when the mediator is entered.

(3) If the IV drops at all it is mediation, but to confirm that the mediation is significant, you generally also conduct and report a significant Sobel test.   (The block where the mediator is entered need not increase the R2 – in fact, if there is an increase in R2 it shows how there is at least some variance in the mediator that is not linked to the distal IV.)  
If the IV drops from a sig beta to an ns beta, that is full mediation.  If it drops from a sig beta to a smaller sig beta that is partial mediation.
As an alternative to a significant Sobel test, recently it has been established in some social psych journals that you can report the results of bootstrapping analysis.  Particularly with small samples, this is v handy, as a marginal Sobel will often be sig with bootstrapping.  (Obviously, if you first do a Sobel and then go to a bootstrapping analysis to chase significance, you increase the likelihood of Type I error (false positives).  Also NB bootstrapping increases the impact of outliers – in some circs this can mean reduced power; in others sig results that won’t replicate.)

How to do this in SPSS

1. Look at the variables

2. Center the continuous variables and recode categorical so zero is meaningful.

3. Look at correlation of IV and mediator and/or run regression predicting mediator from IV and any control variables.

4. Run HMR with IV and control variables in Block 1, mediator in Block 2.

5. Calculate Sobel test online or via WIMP excel file.

6. Create the figure.

1. Use analyse > descriptive > frequencies to get descriptive statistics and histograms for the data.  Have a look for errors and violations of assumptions.  Never skip this step.  Like all regressions, mediation analysis is sensitive to outliers and excessive skew/kurtosis.
FREQUENCIES

  VARIABLES=iv mediator control1 control2 gender group dv1 dv2

  /STATISTICS=STDDEV MINIMUM MAXIMUM SEMEAN MEAN MEDIAN SKEWNESS  SESKEW

  KURTOSIS SEKURT

  /HISTOGRAM

  /ORDER=  ANALYSIS .

2. Check out the inter-correlations among the IVs now and save yourself some trouble.  Your IV should be correlated with the mediator, both should be correlated with the DVs, and your other IVs including control variables should not be correlated.  You can use this syntax to create Table 1.
Under some circs your zero-order correlations need not be sig.  If the correlations are ns, but if you get significant regression coefficients in the same direction as the zero-order correlation for the IV -> Mediator, IV->DV, and Mediator->DV in regression when other variables are controlled for, you’re still ok.  Just means that heaps of variance in DV is linked to controls and needs to be accounted for before the effects of IV and/or Mediator are detectable.
1. Analyze > Correlate > Bivariate

2. enter all ivs and DVs

3. click options > “Exclude cases listwise” and in the same window “Means and standard deviations” > continue

4. click paste

CORRELATIONS

  /VARIABLES= iv mediator control1 control2 gender group dv1 dv2

  /PRINT=TWOTAIL NOSIG

  /STATISTICS DESCRIPTIVES

  /MISSING=LISTWISE .

Run this syntax.  Use the means and standard deviations and inter-correlations to form in Table 1.  Often Table 1 also contains the scale reliabilities in the diagonal.  You get this from earlier reliability analyses when you created the scales.

NB A rule of thumb is anything over .3 you should ponder whether there’s mediation happening or whether the two IVs are tapping the same thing & could be averaged.  See Tabachnick and Fidell (1996) on this point.

3.  Centering and recoding for meaningful zeroes is optional for mediation but a good habit to get into.  It increases the interpretability of coefficients and constant in regression.

Calculate centered scores for all IVs by subtracting the mean:  I like to use c_ as a prefix indicating it’s a centered score.  Work in the syntax window (too much time otherwise going through compute).

iv mediator control1 control2 gender group dv1 dv2

Compute c_iv = iv – [numerical mean as seen in output for correlations or freq above].

Compute c_med = mediator – [numerical mean as seen in output].

Compute c_cont1 = control1 – [numerical mean as seen in output].

Compute c_cont2 = control2 – [numerical mean as seen in output].

execute.

*recode the categorical variables so that they have meaningful zero points and only two levels.  I do not recommend using 1, 2; this has a bad effect on the constant / graphs etc.  Do not use 0, 1 unless the zero group is a baseline or reference group.  I recommend 1, -1 unless you have thought deeply about alternatives.  But if you have extremely unequal n in the two levels you probably should think deeply about alternatives and go with weighted effect coding (e.g., for 75% women, women = +.25 and men -.75).  See Aiken and West (1991) on this point.

If (gender=2) women = 1 .

If (gender=1) women = -1 .

Execute.

*assuming the original coding was women are 2, men 1, this creates a two-group *categorical IV where +1 are women and -1 are men.

For our group variable, if there are 3 groups, we need to create two (k-1) variables for the regression.  

*The first one:

If (group=1) grp1vs23 = 2 .

If (group>1) grp1vs23 = -1 .

Execute.

*creates a contrast code comparing the first group (e.g., a control condition) to the last two.  Another way of doing the same thing is:

If (group=1) grp1vs23 = 2 .

If (group=2) grp1vs23 = -1 .

If (group=3) grp1vs23 = -1 .
Syntax for the second contrast code:

If (group=1) grp2vs3 = 0 .

If (group=2) grp2vs3 = 1 .

If (group=3) grp2vs3 = -1 .

*creates a contrast code comparing the latter two groups to each other.

Execute .

*you pick contrasts that are orthogonal to each other and based on theory.

If there is one meaningful baseline or reference group such as a control condition, you can use dummy coding (0,1) to compare each condition to the controls:

If (group=1) dum2v1 = 0 .

If (group=1) dum3v1 = 0 .

If (group=2) dum2v1 = 1 .

If (group=2) dum3v1 = 0 .

If (group=3) dum2v1 = 0 .

If (group=3) dum3v1 = 1 .
Execute.
*Usually dummy codes less useful than contrast codes in my opinion.

FREQUENCIES

  VARIABLES=c_iv1 c_iv2 c_iv3  women grp1vs23 grp2vs3 

  /STATISTICS=STDDEV MINIMUM MAXIMUM SEMEAN MEAN MEDIAN SKEWNESS SESKEW

  KURTOSIS SEKURT

  /HISTOGRAM

  /ORDER=  ANALYSIS .

*Always check your newly created variables to see if they have reasonable (near zero) means and standard deviations.
You don’t center the DVs as this serves no statistical purpose.  (However, if you do center the DVs, nothing bad happens – you get the exact same regression results.  That is, the R2 and coefficients are the same; only the constant changes.)
3.  Establish relationship of IV and mediator.

****ONE IV, ONE MEDIATOR

Check out whether the IV is correlated with the mediator in Table 1 (or the output from correlations above).  If it is, if your model is simple (e.g., 1 IV, 1 mediator) you can get away with reporting the zero-order correlation. “To test for mediation, it was established that the IV was associated with the mediator, r=, p=, and then a hierarchical multiple regression was conducted with the IV in Block 1, and the mediator in Block 2.”  However, if you have other variables included as controls in the model, NB that a full mediation analysis should control for the effects of any other variables when assessing the link of the IV -> the mediator, not just in the regression with the DV (i.e., you would report two regression analyses, not just a correlation and then regression predicting the DV).  
Moreover, you need to run a regression predicting the mediator by the IV (and any controls) to calculate the Sobel test.  In this case run:
REGRESSION

  /DESCRIPTIVES MEAN STDDEV CORR SIG N

  /MISSING LISTWISE

  /STATISTICS COEFF OUTS R ANOVA COLLIN TOL CHANGE ZPP

  /CRITERIA=PIN(.05) POUT(.10)

  /NOORIGIN

  /DEPENDENT mediator

  /METHOD=ENTER c_iv c_cont1 c_cont2 women grp1vs23 grp2vs3

  /SCATTERPLOT=(*ZRESID ,*ZPRED )

  /RESIDUALS DURBIN HIST(ZRESID) NORM(ZRESID)

  /CASEWISE PLOT(ZRESID) OUTLIERS(3) .

Report e.g. “The IV and the control variables accounted for significant variance in the mediator, R2=, F=, p=.  As expected, the coefficient for the IV was significant, beta=, p=.”
4.  Then you conduct a hierarchical multiple regression with the IV (and controls) in Block 1, along with other control variables, and the mediator in Block 2.
REGRESSION

  /DESCRIPTIVES MEAN STDDEV CORR SIG N

  /MISSING LISTWISE

  /STATISTICS COEFF OUTS R ANOVA COLLIN TOL CHANGE ZPP

  /CRITERIA=PIN(.05) POUT(.10)

  /NOORIGIN

  /DEPENDENT dv1

  /METHOD=ENTER c_iv c_cont1 c_cont2 women grp1vs23 grp2vs3 

  /METHOD=ENTER c_med

  /SCATTERPLOT=(*ZRESID ,*ZPRED )

  /RESIDUALS DURBIN HIST(ZRESID) NORM(ZRESID)

  /CASEWISE PLOT(ZRESID) OUTLIERS(3) .

*Table 2 is created based on the output from this main analysis.  You also report the R2 change for each block and note the significant coefficients in the text.  
E.g., “The IV and the control variables accounted for sig variance in the DV, R2=, F=, and the coefficient for the IV was significant, beta=, p=. [Also report other coefficients.]  In Block 2, the mediator did or did not add sig to the variance accounted for in the DV, R2 ch = , F=.  The coefficient for the mediator was significant, beta=, p=.  When the mediator was entered in Block 2, the coefficient for the IV decreased to beta=,p=.”
5. Then you calculate a Sobel test on whether this decrease is significant/reliable.  I usually do this online at the great web site established by Kris Preacher; briefly run with Geoffrey Leonardelli, and now online at Preacher’s web site again: 
http://www.people.ku.edu/~preacher/sobel/sobel.htm 
NB the web site http://www.people.ku.edu/~preacher/ contains many other interesting and useful links and stats pages.  Just be aware that Kris Preacher is a cutting-edge statistician, not a conventional one.  That means that often times he is proposing sensible but not widely accepted changes to formulae or reporting standards.  Use the ‘standard’ techniques for your thesis and manuscripts unless you have the confidence and knowledge to explain your deviance to your supervisor and reviewers/editors.
And a corollary: Do not pursue ‘correct’ statistical techniques just because a wise person points out the flaws in what you are currently doing.  Often a postgrad is wiser to choose a conventional technique and leave statistical innovation to risk-takers with tenure-track jobs.  Statistical innovation risks leading to irate comments from old school reviewers/editors angry @ implication that once again they will have to learn another $!&@#*% novel statistic.  No one enjoys feeling stupid and out-of-date.  Alternative strategies include reporting the results of the conventional test and footnoting the advanced/innovative test; this is often what I do.  Again, only be innovative if you are sure you know what you’re doing.
If you do not have access to the web you can calculate mediation tests with the excel file WIMP (Winnifred’s Mediation Program).  Or by hand for that matter – the formulae are:
Sobel’s z = a*b/SQRT(b2*sa2 + a2*sb2)

Aroian’s z = a*b/SQRT(b2*sa2 + a2*sb2 + sa2*sb2)

Goodman’s z = a*b/SQRT(b2*sa2 + a2*sb2 - sa2*sb2)

In these formulae, a is the unstandardised (b, not beta) coefficient predicting the mediator by the IV (with all controls in the equation), sa is the standard error of a (printed in the SPSS output next to the b and before the beta).

 b is the unstandardised (b not beta) coefficient from the final block of the second regression predicting the DV by the mediator (with the IV and all other controls in the equation).
NB these #s come from two separate regressions – the first two come from the regression predicting the mediator from the IV with all controls in the equation.  The second two come from the regression predicting the DV from the IV in block 1 and the mediator in block 2 – you take the #s from the coefficients for mediator in Block 2.

Bootstrapping


As noted above, as an alternative to a significant Sobel test, recently it has been established in some social psych journals that you can report the results of bootstrapping analysis.  Particularly with small samples, this is v handy, as a marginal Sobel will often be sig with bootstrapping.  (Obviously, if you first do a Sobel and then go to a bootstrapping analysis to chase significance, you increase the likelihood of Type I error (false positives).  Also NB bootstrapping increases the impact of outliers – in some circs this can mean reduced power.)

To justify using bootstrapping on small samples you can cite:

Preacher, K. J., & Hayes, A. F. (2004). SPSS and SAS procedures for estimating indirect effects in simple mediation models. Behavior Research Methods, Instruments, & Computers, 36(4), 717-731. [online http://www.people.ku.edu/~preacher/sobel/preacher_hayes.pdf ]

To do this in SPSS, you can follow the instructions at the preacher web site.  Or as a shortcut:

1. In the unlikely event you have only an IV and a mediator and no control variables, then with your real data set open and ready to go, download the file:

http://www.comm.ohio-state.edu/ahayes/SPSS%20programs/sobel_spss.SPS .

Save it to your computer and open in SPSS.  Highlight and run the entire file without changing anything (i.e., control A, control R).


Then in a new syntax file which you can save as a record of what you were doing, type:
SOBEL y=yvar/x=xvar/m=mvar/boot=z.

where yvar is the name of the dependent variable in your open data file, xvar is the name of the independent variable, mvar is the name of the proposed mediating variable, and z specifies the number of bootstrap resamples desired, in increments of 1000 up to a maximum of 1,000,000. For example, if z is set to 3000, then the bootstrap estimates will be based on 3000 resamples. If z is set to 0 (or any number less than 1000), the bootstrapping module is deactivated. 
[From the website, http://www.comm.ohio-state.edu/ahayes/sobel.htm ]”
This will output the Sobel test, which should be the same as what you get if you do the Sobel normally, as well as the ‘BOOTSTRAP RESULTS FOR INDIRECT EFFECT’.  You report the effect, s.e., and the 95% confidence intervals.  If these do not include zero, your indirect effect is sig @ p < .05. 
The write-up could be something like this (quoting from Barlow, Louis & Pedersen, ’08, which in turn we adapted from an ms of Bill von H’s; feel free to paraphrase):

As recommended for small samples, we used nonparametric bootstrapping

analyses (see Preacher & Hayes, 2004; Preacher, Rucker, & Hayes, 2007) to test the meditational model of moral outrage as a mediator of the relationship between false beliefs and attitudes towards Aboriginal Australians. In these analyses, mediation is significant if the 95% Bias Corrected and accelerated confidence intervals for the indirect effect do not include 0 (Preacher & Hayes, 2004; Preacher et al., 2007). Results based on 10000 bootstrapped samples indicated that whilst the total effect of false beliefs on attitudes towards Aboriginal Australians was significant (TE=-2.68, SE=1.30, p=.041),

the direct effect was not (DE=-.80, SE=1.18, p=.498). Moral outrage fully mediated the relationship between false beliefs and attitudes towards Aboriginal Australians (IE lower 95% CI=-3.56, upper 95% CI=-.51), such participants who indicated high levels of false beliefs were less likely to feel moral outrage, and through low levels of moral outrage, less likely to indicate positivity about Aboriginal Australians. Because zero is not in the 95% confidence interval, the indirect effect is significantly different from zero at p < .05 (two tailed).
(BTW if the residual direct effect were significant as well as the indirect effect, that would indicate partial mediation.)

Bootstrapping is relatively new, so don’t be surprised if they make you footnote the Sobel anyways, or just reject the statistic altogether.
2. A more likely event is that you have other control variables as well that you want to consider.  Then download the file:

http://www.comm.ohio-state.edu/ahayes/SPSS%20programs/indirect.SPS from the website http://www.comm.ohio-state.edu/ahayes/SPSS%20programs/indirect.htm  .

Save it to your computer and open and run the entire file without changing anything (i.e., control A, control R).


Then in a new syntax file which you can save as a record of what you were doing, type:”

INDIRECT y = dv/x = iv/m = mlist covlist/c = cov/boot = z/conf = ci/
         /normal = n/contrast = t/percent = p/bc = b/bca = d
 
where dv is the name of the dependent variable…, iv is the name of the independent variable …, and mlist is a list of mediator variables ….  These are the only arguments that are necessary for execution of the macro.  If no other options are provided, the macro will estimate the paths in the model (using OLS) assuming no control variables, the number of bootstraps will be set to 1000, the confidence level will default to 95, only bias corrected and adjusted confidence intervals for the indirect effects will be printed, and no normal theory results or pairwise contrasts are conducted.  For more options, the following arguments are used:

 

covlist is a list of covariate variables, and cov is the number of covariate variables in the covlist list, z is the number of desired bootstrap resamples desired in increments of 1000 (e.g., boot = 2000 yields 2000 bootstrap resamples; set it to 0 to disable bootstrapping), ci is the desired confidence level, p is set to 1 to print percentile-based confidence intervals, b is set to 1 to print bias-corrected confidence intervals, d is set to 0 to disable printing of bias corrected and adjusted confidence intervals, n is set to 1 to enable printing of normal theory test results, and t is set to 1 to do all possible pairwise contrasts between indirect effects.  If any of these arguments are not provided, default values will be used (the defaults are c = 0, z = 1000, ci = 95, p = 0, b =0, d = 1, n = 0, t = 0).  Note that if c is set to 0 it is assumed that there are no variables listed in covlist, and all variables listed after “m =” are treated as potential mediators.  

If covariates are provided, normal theory results will not be printed. “

 [From website http://www.comm.ohio-state.edu/ahayes/SPSS%20programs/indirect.htm ]

Use the same write-up details as above, only add a clause about the covariates:

“Mediation analyses based on 3000 bootstrapped samples using bias-corrected and accelerated 95% confidence intervals (Preacher & Hayes, 2004) showed that controlling for the effect of covariate 1 [b=, se=, p=] and covariate 2, [b=, se=, p=], the IV had a significant total effect on the DV [TE=, se=, p=], a significant residual direct effect [DE=, se=, p=], and a significant indirect effect [IE=, se=, LL=, UL=].”
LL and UL above are acronyms for the lower and upper level confidence intervals.  The macro online outputs bias-corrected and accelerated confidence intervals – sounds good doesn’t it?  (This means the significance testing is ostensibly more robust in the face of distribution issues common to small samples.  As noted above, I think the problem is more about Type 1 than 2 errors w/ this particular statistic …).   

Just a reminder: if the range from the LL to the UL doesn’t include zero, your indirect effect is significant.
 

6.  Create the figure (almost always in theses; less often in manuscripts, but figures are expensive so these can be dropped).  It usually looks like this:




Note. * p < .05 ** p < .01, *** p < .001
Where βa = the beta coefficient (standardized – not the b or unstandardised that we used for the Sobel test) of the IV -> the mediator (with all controls in the equation), βb is the beta for the mediator -> DV with IV and controls in the equation, βc is the coefficient for the IV when the mediator and controls are in the equation, and βd is the coefficient for the IV when the controls are in the equation but the mediator has not been entered.  Usually each coeffient is marked * to indicate p < .05, ** for p < .01, etc..

With bootstrapping you can do the same type of figure – just clearly describe in the note what the coefficients are so people don’t freak out @ your small betas.

*****

About Suppression

[CONFUSION ALERT – you can skip the next 4 paragraphs if you don’t want to hurt your brain thinking about suppression models now.] What about if you get a sig coefficient for IV-> DV in the OPPOSITE direction to the ns zero-order correlation, when another variable is controlled?  This suggests a suppression relationship; the ‘mediator(s)’ is called the ‘suppressor(s)’.  It has two completely different interpretations.

In one case, the IV is having a direct effect on the DV in one direction at the same time as it is having an indirect effect on the DV in the other direction.  For example, in Nyla Branscombe’s rejection-identification model, discrimination has a direct negative impact on well-being at the same time as it has an indirect positive impact via increased identification.  (On the one hand, discrimination is harmful to wellbeing.  On the other hand, it discrimination boosts identification, and that has a positive (buffering) impact.  So the overall zero-order correlation could be ns, but significant coefficients appear from the IV when the mediator is controlled and a significant indirect effect in the opposite direction is found if you do meditational analyses.)  You propose such a model based on theory and you test it by testing the significance of the indirect effect with a Sobel or bootstrapping.  Unlike normal mediation analysis you don’t have to show the IVpredicts the DV in Block 1 but you still need to show the IV predicts the suppressor and there is a sig indirect effect of IV -> DV via the suppressor.

Be aware that theoretically meaningful suppression effects are perhaps less common than the other interpretation –  statistical artefacts which result from using inter-correlated IVs whose relationship is not ‘outside’ the theoretical model.  Multiple regression assumes that the IVs are uncorrelated.  If they are correlated and the relationship among the IVs is irrelevant to the relationships of each predictor to the DV, one IV’s effect may be exaggerated [the more reliable one] and the other is driven in the opposite direction to compensate – sometimes producing as a spurious suppression effect.  If the overlap between the IVs drives the relationships of the predictors to the DV (e.g., if you include two measures tapping the same latent construct, which is associated with the DV) the residual unique effects may be under-estimated.  

Put another way, correlated IVs are appropriate for standard multiple regression analyses ONLY if the researcher is prepared to accept the relationships among the IVs as a given (leaving aside the question of whether A caused B, vice versa, or there is a third factor) and to assume that the relationships among the IVs are unrelated to the relationships of the IVs independently to the DV; in my view this is rarely a safe assumption.  If the IV and the suppressor are correlated, and if their common variance is unrelated to the prediction of the DV, then when the suppressor is controlled, the relationship between the IV and the DV is artificially “boosted” in the opposite direction to the real IV-DV relationship as reflected in the zero-order correlation.  Sound worrying?  It is!  You only interpret suppression effects as meaningful if there is a good theoretical reason to believe that the IV has direct and indirect effects on the DV in opposing directions.
******************************************************

3.  Establish relationship of IV and mediator.

****TWO IVs, ONE MEDIATOR

The bootstrapping syntax above will also give you some output with multiple mediators and IVs, I think, but if you want to keep to regression analyses you can do that too.  You’re still doing 2 regression analyses, but now are reporting two sets of coefficients that we are interested in, and two Sobel tests.

Imagine our c_iv plus another c_iv2 

REGRESSION

  /DESCRIPTIVES MEAN STDDEV CORR SIG N

  /MISSING LISTWISE

  /STATISTICS COEFF OUTS R ANOVA COLLIN TOL CHANGE ZPP

  /CRITERIA=PIN(.05) POUT(.10)

  /NOORIGIN

  /DEPENDENT mediator

  /METHOD=ENTER c_iv c_iv2 c_cont1 c_cont2 women grp1vs23 grp2vs3

  /SCATTERPLOT=(*ZRESID ,*ZPRED )

  /RESIDUALS DURBIN HIST(ZRESID) NORM(ZRESID)

  /CASEWISE PLOT(ZRESID) OUTLIERS(3) .

Report e.g. “The ivs and the control variables accounted for sig variance in the mediator, R2=, F= .  The coefficient for IV1 was significant, beta=, p=, and so was the coefficient for IV2, beta=,p=.” [ie both contribute independently to predict mediator]
4.  Then you conduct a hierarchical multiple regression with the IVs (and controls) in Block 1, and the mediator in Block 2.
REGRESSION

  /DESCRIPTIVES MEAN STDDEV CORR SIG N

  /MISSING LISTWISE

  /STATISTICS COEFF OUTS R ANOVA COLLIN TOL CHANGE ZPP

  /CRITERIA=PIN(.05) POUT(.10)

  /NOORIGIN

  /DEPENDENT dv1

  /METHOD=ENTER c_iv c_iv2 c_cont1 c_cont2 women grp1vs23 grp2vs3 

  /METHOD=ENTER c_med

  /SCATTERPLOT=(*ZRESID ,*ZPRED )

  /RESIDUALS DURBIN HIST(ZRESID) NORM(ZRESID)

  /CASEWISE PLOT(ZRESID) OUTLIERS(3) .

*Table 2 is created based on the output from this main analysis.  You also report the R2 change for each block and note the significant coefficients in the text.  Report both IVs sig in b1 and decreasing in b2.
E.g., “The IV and the control variables accounted for sig variance in the DV, R2=, F=, and the coefficient for the IV was significant, beta=, p=. , and so was the coefficient for IV2.  [May also report other control coefficients.]  In Block 2, the mediator did or did not add sig to the variance accounted for in the DV, R2 ch = , F=.  The coefficient for the mediator was significant, beta=, p=.  When the mediator was entered in Block 2, the coefficient for the IV decreased to beta=,p=, z= , and the coefficient for IV2 decreased to beta=,p=., z=”
5. Then you calculate two Sobel tests for whether each of these decreases is significant/reliable, using web site, WIMP file, or formulae above.
6.  Create figures to depict this.
****MORE THAN ONE MEDIATOR

*at this stage start thinking about path analysis using EQS or AMOS, etc..  But it can be done in regression.

3.  Shortest possible analysis: Need one regression analysis for each potential mediator plus one regression analysis for DV. [Longer analysis discussed below – one regression per mediator plus one on DV for each mediator.]
Imagine our c_iv plus another c_iv2, and c_med plus another med2 [c_med2].  Each IV has to predict each mediator controlling for the other IVs and the other mediator.

REGRESSION

  /DESCRIPTIVES MEAN STDDEV CORR SIG N

  /MISSING LISTWISE

  /STATISTICS COEFF OUTS R ANOVA COLLIN TOL CHANGE ZPP

  /CRITERIA=PIN(.05) POUT(.10)

  /NOORIGIN

  /DEPENDENT mediator

  /METHOD=ENTER c_iv c_iv2 c_med2 c_cont1 c_cont2 women grp1vs23 grp2vs3

  /SCATTERPLOT=(*ZRESID ,*ZPRED )

  /RESIDUALS DURBIN HIST(ZRESID) NORM(ZRESID)

  /CASEWISE PLOT(ZRESID) OUTLIERS(3) .

REGRESSION

  /DESCRIPTIVES MEAN STDDEV CORR SIG N

  /MISSING LISTWISE

  /STATISTICS COEFF OUTS R ANOVA COLLIN TOL CHANGE ZPP

  /CRITERIA=PIN(.05) POUT(.10)

  /NOORIGIN

  /DEPENDENT med2

  /METHOD=ENTER c_iv c_iv2 c_med c_cont1 c_cont2 women grp1vs23 grp2vs3

  /SCATTERPLOT=(*ZRESID ,*ZPRED )

  /RESIDUALS DURBIN HIST(ZRESID) NORM(ZRESID)

  /CASEWISE PLOT(ZRESID) OUTLIERS(3) .

Report e.g. “Controlling for med2, the ivs and the control variables accounted for sig variance in mediator, R2=, F= .  The coefficient for IV1 was significant, beta=, p=, and so was the coefficient for IV2, beta=,p=.” [ie both contribute independently to predict mediator, even when med2 is controlled].  In addition, controlling for mediator, the ivs and the control variables accounted for sig variance in med2, R2=, F= .  The coefficient for IV1 was significant, beta=, p=, and so was the coefficient for IV2, beta=,p=.” [ie both contribute independently to predict med2, even when mediator is controlled].  
4.  Then you conduct a hierarchical multiple regression with the IVs (and controls) in Block 1, and the mediators in Block 2.
REGRESSION

  /DESCRIPTIVES MEAN STDDEV CORR SIG N

  /MISSING LISTWISE

  /STATISTICS COEFF OUTS R ANOVA COLLIN TOL CHANGE ZPP

  /CRITERIA=PIN(.05) POUT(.10)

  /NOORIGIN

  /DEPENDENT dv1

  /METHOD=ENTER c_iv c_iv2 c_cont1 c_cont2 women grp1vs23 grp2vs3 

  /METHOD=ENTER c_med c_med2

  /SCATTERPLOT=(*ZRESID ,*ZPRED )

  /RESIDUALS DURBIN HIST(ZRESID) NORM(ZRESID)

  /CASEWISE PLOT(ZRESID) OUTLIERS(3) .

*Table 2 is created based on the output from this main analysis.  You also report the R2 change for each block and note the significant coefficients in the text.  Report both IVs sig in b1 and decreasing in b2.  Report both meds sig in B2.
E.g., “The IV and the control variables accounted for sig variance in the DV, R2=, F=, and the coefficient for the IV was significant, beta=, p=. , and so was the coefficient for IV2.  [May also report other control coefficients.]  In Block 2, the mediators did or did not add sig to the variance accounted for in the DV, R2 ch = , F=.  The coefficient for the mediator was significant, beta=, p=, as was the coefficient for med2, beta=,p=.  In Block 2, the coefficient for the IV decreased to beta=,p=, and Sobel tests in which the other mediator was controlled revealed indirect effects via mediator, z=,p=, and via med2, z=,p=.  The coefficient for IV2 decreased in Block 2 to beta=,p=., and again, significant mediation was observed via mediator, z=, p=, and via med2, z=,p=”
5. Then you calculate FOUR Sobel tests for whether each of these decreases is significant/reliable, using web site, WIMP file, or formulae above.  Plug the zs into the write-up above.
6.  Create figures to depict this.

NB: The above analysis is shorter and simpler than the one that is technically required, but since it is conservative (ie, any sig Sobel you find can be trusted) I would go with the above analysis most of the time.

Technically the Sobel test that you would calculate using the above data does not compare the coefficients for IV in block1 with block 2, it compares the coefficient for IV on the DV accounting for the other mediator.  If you wanted to write up a more complex but correct report you would need to do separate regression analyses on the DV for each mediational path.

REGRESSION

  /DESCRIPTIVES MEAN STDDEV CORR SIG N

  /MISSING LISTWISE

  /STATISTICS COEFF OUTS R ANOVA COLLIN TOL CHANGE ZPP

  /CRITERIA=PIN(.05) POUT(.10)

  /NOORIGIN

  /DEPENDENT dv1

  /METHOD=ENTER c_iv c_iv2 c_med2 c_cont1 c_cont2 women grp1vs23 grp2vs3 

  /METHOD=ENTER c_med 

  /SCATTERPLOT=(*ZRESID ,*ZPRED )

  /RESIDUALS DURBIN HIST(ZRESID) NORM(ZRESID)

  /CASEWISE PLOT(ZRESID) OUTLIERS(3) .

E.g., “Controlling for med2, the IV and the control variables accounted for sig variance in the DV, R2=, F=.  The coefficient for the IV was significant, beta=, p=. , and so was the coefficient for IV2.  [May also report other control coefficients and med2.]  In Block 2, the mediator did or did not add sig to the variance accounted for in the DV, R2 ch = , F=.  The coefficient for the mediator was significant, beta=, p=, and the coefficient for the IV decreased to beta=,p=, z=,p=, while the coefficient for IV2 decreased to beta=,p=, z=,p=, showing significant mediation via mediator in both cases.  A second analysis showed that controlling for mediator the IV and the control variables accounted for sig variance in the DV, R2=, F=, and the coefficients were significant for iv, beta=,p=, and iv2, beta=,p=.  [May also report other control coefficients and mediator.]  In Block 2, the addition of med2 did or did not add sig to the variance accounted for in the DV, R2 ch = , F=.  The coefficient for the IV decreased to beta=,p=, z=,p=, while the coefficient for IV2 decreased to beta=,p=, z=,p=, showing significant mediation via med2 in both cases.”

This analysis is too complex for Tables – you would just skip Table 2 or create a version where the coefficients are shown only for the final model, which is the same in each of the analyses.  The entry block coefficients differ depending on which mediator is being controlled.

